
Gramatyki metamorficzne
Gramatyki bezkontekstowe

I Gramatykπ bezkontekstowπ nazywamy uporzπdkowanπ
czwórkÍ G = È�, N, S, PÍ, gdzie
I � jest skoÒczonym i niepustym alfabetem (symbole terminalne)
I N jest skoÒczonym i niepustym zbiorem symboli

nieterminalnych
I S œ N jest wyróønionym nieterminalem nazywanym symbolem

poczπtkowym
I P jest skoÒczonym zbiorem regu≥ produkcji postaci n æ w ,

gdzie n œ N oraz w œ (� fi N)ú

I Jeúli u = uÕnuÕÕ, dla uÕ, uÕÕ œ (� fi N)ú, oraz n æ w œ P, to
mówimy, øe s≥owo v = uÕwuÕÕ jest w jednym kroku
wyprowadzalne ze s≥owa u, co zapisujemy u ∆ v .

Gramatyki metamorficzne
Gramatyki bezkontekstowe

I Niech ú∆ bÍdzie domkniÍciem tranzytywnym relacji
wyprowadzenia ∆.

I Mówimy, øe s≥owo w œ �ú jest wyprowadzalne w gramatyce
G , jeúli S ú∆ w .

I JÍzyk wszystkich s≥ów wyprowadzalnych w gramatyce G
oznaczamy przez L(G):

L(G) = {w œ �ú|S ú∆ w}.

I Symbolem Á bÍdziemy oznaczaÊ s≥owo puste.

Gramatyki metamorficzne
Gramatyki bezkontekstowe

Example (JÍzyk palindromów)
Niech G = È�, N, S, PÍ, gdzie
I � = {a, b}
I N = {S}
I P = {S æ Á, S æ a, S æ b, S æ aSa, S æ bSb}

Wówczas jÍzyk L(G) sk≥ada siÍ z palindromów nad alfabetem
{a, b}:

L(G) = {Á, a, b, aa, bb, aaa, aba, bab, bbb, aaaa, abba, baab, bbbb, . . .}



Gramatyki metamorficzne
Gramatyki bezkontekstowe

Example (JÍzyk poprawnie rozstawionych nawiasów)
Niech G = È�, N, S, PÍ, gdzie
I � = {(, )}
I N = {S}
I P = {S æ Á, S æ (S)S}

Wówczas jÍzyk L(G) sk≥ada siÍ ze s≥ów w których poprawnie
rozstawiono nawiasy:

L(G) = {Á, (), ()(), (()), ()()(), ()(()), (())(), (()()), ((())), . . .}

Gramatyki metamorficzne
Gramatyki bezkontekstowe

Example (Prosta gramatyka zdaÒ jÍzyka angielskiego)
Niech G = È�, N, zdanie, PÍ, gdzie
I � = {the, apple, man, eats, sings}
I

N =

Y
_______]

_______[

zdanie,
fraza_rzecz ,
fraza_czas,
przedimek,
rzeczownik,
czasownik

Z
_______̂

_______\

Gramatyki metamorficzne
Gramatyki bezkontekstowe

Example (cd.)
I

P =

Y
______________]

______________[

zdanie æ fraza_rzecz fraza_czas,
fraza_rzecz æ przedimek rzeczownik,
fraza_czas æ czasownik fraza_rzecz ,
fraza_czas æ czasownik,
przedimek æ the,
rzeczownik æ apple,
rzeczownik æ man,
czasownik æ eats,
czasownik æ sings

Z
______________̂

______________\

Gramatyki metamorficzne
Gramatyki bezkontekstowe

Example (cd.)
Wówczas jÍzyk L(G) sk≥ada siÍ z kilkunastu prostych zdaÒ jÍzyka
angielskiego:

L(G) =

Y
_____________________]

_____________________[

the apple eats the apple,
the apple eats the man,
the apple sings the apple,
the apple sings the man,
the apple eats,
the apple sings,
the man eats the apple,
the man eats the man,
the man sings the apple,
the man sings the man.
the man eats,
the man sings

Z
_____________________̂

_____________________\



Gramatyki metamorficzne
Analiza sk≥adniowa w Prologu

Spróbujmy napisaÊ analizator w postaci predykatów akceptujπcych
listy s≥ów:

zdanie(X) :-

append(Y, Z, X),

fraza_rzecz(Y), fraza_czas(Z).

fraza_rzecz(X) :-

append(Y, Z, X),

przedimek(Y), rzeczownik(Z).

fraza_czas(X) :-

append(Y, Z, X),

czasownik(Y), fraza_rzecz(Z).

fraza_czas(X) :-

czasownik(X).

Gramatyki metamorficzne
Analiza sk≥adniowa w Prologu

przedimek([the]).

rzeczownik([apple]).

rzeczownik([man]).

czasownik([eats]).

czasownik([sings]).

Przyk≥ady analizy i syntezy zdaÒ:
?- zdanie([the, man, eats, the, apple]).

true ;

false.

?- zdanie(X).

X = [the, apple, eats, the, apple] ;

X = [the, apple, eats, the, man] ;

ERROR: Out of global stack

Exception: (8) fraza_czas(_G21) ?

Gramatyki metamorficzne
Analiza sk≥adniowa w Prologu

Zapiszemy predykaty w ten sposób aby wydziela≥y w liúcie prefiks
wyprowadzalny z danego nieterminala i oddawa≥y pozosta≥y sufiks
listy:

zdanie(X, Z) :-

fraza_rzecz(X, Y), fraza_czas(Y, Z).

fraza_rzecz(X, Z) :-

przedimek(X, Y), rzeczownik(Y, Z).

fraza_czas(X, Z) :-

czasownik(X, Y), fraza_rzecz(Y, Z).

fraza_czas(X, Y) :-

czasownik(X, Y).

Gramatyki metamorficzne
Analiza sk≥adniowa w Prologu

przedimek([the | X], X).

rzeczownik([apple | X], X).

rzeczownik([man | X], X).

czasownik([eats | X], X).

czasownik([sings | X], X).

Przyk≥ad syntezy zdaÒ:
?- zdanie(X, []).

X = [the, apple, eats, the, apple] ;

X = [the, apple, eats, the, man] ;

X = [the, apple, sings, the, apple] ;

X = [the, apple, sings, the, man] ;

X = [the, apple, eats] ; X = [the, apple, sings] ;

X = [the, man, eats, the, apple] ;

X = [the, man, eats, the, man] ;

X = [the, man, sings, the, apple] ;

X = [the, man, sings, the, man] ;

X = [the, man, eats] ; X = [the, man, sings].



Gramatyki metamorficzne
Produkcje gramatyki jako klauzule

W Prologu produkcje gramatyki moøna zapisywaÊ wprost w
programie:

zdanie --> fraza_rzecz, fraza_czas.

fraza_rzecz --> przedimek, rzeczownik.

fraza_czas --> czasownik, fraza_rzecz.

fraza_czas --> czasownik.

przedimek --> [the].

rzeczownik --> [apple].

rzeczownik --> [man].

czasownik --> [eats].

czasownik --> [sings].

Symbole nieterminalne zapisuje siÍ w takiej postaci jak nazwy
predykatów, symbole terminalne podaje siÍ miÍdzy nawiasami
kwadratowymi, strza≥kÍ zapisuje siÍ trzema znakami -->.

Gramatyki metamorficzne
Produkcje gramatyki jako klauzule

Jeøeli ten sam nieterminal wystÍpuje w wiÍcej niø jednej produkcji,
to prawe ich strony moøna po≥πczyÊ úrednikiem (alternatywa):

zdanie --> fraza_rzecz, fraza_czas.

fraza_rzecz --> przedimek, rzeczownik.

fraza_czas --> czasownik, fraza_rzecz; czasownik.

przedimek --> [the].

rzeczownik --> [apple]; [man].

czasownik --> [eats]; [sings].

Gramatyki metamorficzne
Produkcje gramatyki jako klauzule

Do analizy i syntezy s≥uøπ predykaty phrase/2 i phrase/3.
W pierwszym przypadku ma on postaÊ phrase(+Nieterminal,

?ListaTerminali):

?- phrase(zdanie, [the, man, eats, the, apple]).

true .

?- findall(X, phrase(zdanie, X), L).

L = [[the,apple,eats,the,apple],[the,apple,eats,the,man],

[the,apple,sings,the,apple],[the,apple,sings,the,man],

[the,apple,eats],[the,apple,sings],

[the,man,eats,the,apple],[the,man,eats,the,man],

[the,man,sings,the,apple],[the,man,sings,the,man],

[the,man,eats],[the,man,sings]]

Gramatyki metamorficzne
Produkcje gramatyki jako klauzule

W drugim przypadku ma on postaÊ phrase(+Nieterminal,

?ListaTerminali, ?ListaPozosta≥ychTerminali), przy czym
trzecim parametrem odbiera siÍ sufiks, który nie jest wyprowadzony
z podanego nieterminala:

?- phrase(fraza_rzecz, [the, man, eats, the, apple], X).

X = [eats, the, apple].

?- phrase(zdanie, [the, man, eats, the, apple], X).

X = [] .



Gramatyki metamorficzne
Produkcje gramatyki jako klauzule

Example (JÍzyk poprawnie rozstawionych nawiasów)
W SWI-Prologu, poczπwszy od wersji 7, aby ≥aÒcuch znaków
traktowany by≥ w postaci listy kodów naleøy umieúciÊ go miÍdzy
odwróconymi apostrofami (w starszych wersjach miÍdzy
cudzys≥owami).

s2 --> ‘‘.

s2 --> ‘(‘, s2, ‘)‘, s2.

Dialog:

?- phrase(s2, ‘(())‘).

true .

?- phrase(s2, ‘())(‘).

false.

Gramatyki metamorficzne
Parametry nieterminali

W Prologu nieterminale mogπ mieÊ parametry, dziÍki którym nie
tylko mogπ przeprowadzaÊ analizÍ sk≥adniowπ ale równieø
dokonywaÊ przekszta≥cenia sparsowanego s≥owa do postaci
prologowego termu (stπd nazwa gramatyki metamorficzne).
Example (Kody regularnych drzew binarnych)
Regularne drzewa binarne bÍdziemy zapisywaÊ w postaci termów
zbudowanych z funktora f/2 (wierzcho≥ki wewnÍtrzne) i sta≥ej a

(liúcie).
Drzewo kodowaÊ bÍdziemy w postaci ciπgu zer i jedynek w sposób
nastÍpujπcy:
I LiúÊ a kodujemy cyfrπ 0.
I Drzewo f(X, Y) o dwóch poddrzewach X i Y, kodujemy w

postaci ciπgu z≥oøonego kolejno z cyfry 1, nastÍpnie kodu
poddrzewa X i na koÒcu kodu poddrzewa Y.

Gramatyki metamorficzne
Parametry nieterminali

Example (cd.)
Gramatyka metamorficzna dla kodów regularnych drzew binarnych:

kod(a) --> ‘0‘.

kod(f(X, Y)) --> ‘1‘, kod(X), kod(Y).

Dialog:

?- phrase(kod(X), ‘1100100‘).

X = f(f(a, a), f(a, a)) .

?- phrase(kod(f(f(a, a), a)), X), format(’~s~n’, [X]).

11000

X = [49, 49, 48, 48, 48].

Gramatyki metamorficzne
Cele umieszczane w produkcjach

Do produkcji gramatyki moøna wstawiaÊ prologowe cele
umieszczajπc je w nawiasach klamrowych.
Example (Liczby binarne)

bin(X) --> bin(0, X).

bin(X, X) --> ‘‘.

bin(X, Z) --> ‘0‘, {Y is 2*X}, bin(Y, Z).

bin(X, Z) --> ‘1‘, {Y is 2*X+1}, bin(Y, Z).

Dialog:

?- phrase(bin(N), ‘101010‘).

N = 42 .

?- phrase(bin(N), ‘11111111‘).

N = 255 .



Gramatyki metamorficzne
Cele umieszczane w produkcjach

Example (Gramatyka metamorficzna g≥ównych liczebników)

number(0) --> [zero].

number(N) --> xxx(N).

xxx(N) --> digit(D), [hundred], rest_xxx(N1),

{N is D*100 + N1}.

xxx(N) --> xx(N).

rest_xxx(0) --> [].

rest_xxx(N) --> [and], xx(N).

xx(N) --> digit(N).

xx(N) --> teen(N).

xx(N) --> tens(T), rest_xx(N1), {N is T+N1}.

rest_xx(0) --> [].

rest_xx(N) --> digit(N).

Gramatyki metamorficzne
Cele umieszczane w produkcjach

Example (cd.)

digit(1) --> [one]. digit(2) --> [two].

digit(3) --> [three]. digit(4) --> [four].

digit(5) --> [five]. digit(6) --> [six].

digit(7) --> [seven]. digit(8) --> [eight].

digit(9) --> [nine].

teen(10) --> [ten]. teen(11) --> [eleven].

teen(12) --> [twelve]. teen(13) --> [thirteen].

teen(14) --> [fourteen]. teen(15) --> [fifteen].

teen(16) --> [sixteen]. teen(17) --> [seventeen].

teen(18) --> [eighteen]. teen(19) --> [nineteen].

Gramatyki metamorficzne
Cele umieszczane w produkcjach

Example (cd.)

tens(20) --> [twenty]. tens(30) --> [thirty].

tens(40) --> [forty]. tens(50) --> [fifty].

tens(60) --> [sixty]. tens(70) --> [seventy].

tens(80) --> [eighty]. tens(90) --> [ninety].

Dialog:

?- phrase(number(N), [thirteen]).

N = 13 .

?- phrase(number(N), [twenty, five]).

N = 25 .

?- phrase(number(N), [six, hundred, and, sixty, six]).

N = 666 .

Gramatyki metamorficzne
T≥umaczenie zdaÒ na formu≥y rachunku predykatów

I Rzeczownikowi w≥asnemu odpowiada prologowa sta≥a np.
john.

I Rzeczownikowi odpowiada jednoargumentowy predykat
wyraøajπcy w≥asnoúÊ np. man(X).

I Czasownikowi nieprzechodniemu odpowiada
jednoargumentowy predykat wyraøajπcy w≥asnoúÊ np.
lives(X).

I Czasownikowi przechodniemu odpowiada dwuargumentowy
predykat wyraøajπcy relacjÍ np. loves(X, Y).

I Przedimkowi every odpowiada kwantyfikator ogólny.
I Przedimkowi a odpowiada kwantyfikator szczegó≥owy

(egzystencjalny).
I Fraza rzeczownikowa moøe zawieraÊ klauzulÍ wzglÍdnπ

zaczynajπcπ siÍ od s≥owa that, która precyzuje rzeczownik
podajπc jego dodatkowe w≥asnoúci i relacje.



Gramatyki metamorficzne
T≥umaczenie zdaÒ na formu≥y rachunku predykatów

:- op(900, xfx, =>).

:- op(800, xfy, &).

:- op(550, xfy, :).

zdanie(P) -->

fraza_rzecz(X, P1, P), fraza_czas(X, P1).

fraza_rzecz(X, P1, P) -->

przedimek(X, P2, P1, P), rzeczownik(X, P3),

klauzula_wzgl(X, P3, P2).

fraza_rzecz(X, P, P) -->

rzeczownik_wlasny(X).

fraza_czas(X, P) -->

czasownik_przech(X, Y, P1), fraza_rzecz(Y, P1, P).

fraza_czas(X, P) -->

czasownik_nieprzech(X, P).

Gramatyki metamorficzne
T≥umaczenie zdaÒ na formu≥y rachunku predykatów

klauzula_wzgl(X, P1, P1&P2) -->

[that], fraza_czas(X, P2).

klauzula_wzgl(_, P, P) -->

[].

przedimek(X, P1, P2, all(X):(P1=>P2)) -->

[every].

przedimek(X, P1, P2, exists(X):(P1&P2)) -->

[a].

rzeczownik(X, man(X)) -->

[man].

rzeczownik(X, woman(X)) -->

[woman].

Gramatyki metamorficzne
T≥umaczenie zdaÒ na formu≥y rachunku predykatów

rzeczownik_wlasny(john) -->

[john].

czasownik_przech(X, Y, loves(X,Y)) -->

[loves].

czasownik_przech(X, Y, knows(X,Y)) -->

[knows].

czasownik_nieprzech(X, lives(X)) -->

[lives].

Gramatyki metamorficzne
T≥umaczenie zdaÒ na formu≥y rachunku predykatów

Dialog:

?- phrase(zdanie(X), [john, lives]).

X = lives(john).

?- phrase(zdanie(X), [john, knows, john]).

X = knows(john, john) .

?- phrase(zdanie(X), [john, knows, a, woman]).

X = exists(G1): (woman(G1)&knows(john, G1)) .

?- phrase(zdanie(X), [john, loves, every, woman]).

X = all(G1): (woman(G1)=>loves(john, G1)) .

?- phrase(zdanie(X), [every, man, loves, a, woman]).

X = all(G1): (man(G1)=>exists(G2):

(woman(G2)&loves(G1, G2))) .



Gramatyki metamorficzne
T≥umaczenie zdaÒ na formu≥y rachunku predykatów

Dialog:

?- phrase(zdanie(X), [john, loves, every, woman,

that, loves, john]).

X = all(G1): (woman(G1)&loves(G1, john)=>loves(john, G1)) .

’G1(woman(G1) · loves(G1, john) æ loves(john, G1))

?- phrase(zdanie(X), [every, man, loves, every, woman,

that, knows, john]).

X = all(G1): (man(G1)=>all(G2):

(woman(G2)&knows(G2, john)=>loves(G1, G2)))

’G1(man(G1) æ ’G2(woman(G2)·knows(G2, john) æ loves(G1, G2)))

Gramatyki metamorficzne
T≥umaczenie zdaÒ na formu≥y rachunku predykatów

Dialog:

?- phrase(zdanie(X), [john, loves, a, woman,

that, knows, a, man,

that, knows, john]).

X = exists(G1):

((woman(G1)&exists(G2):

((man(G2)&knows(G2, john))&

knows(G1, G2)))&loves(john, G1))

÷G1 (woman(G1) · ÷G2 (man(G2) · knows(G2, john) · knows(G1, G2)) · loves(john, G1))

G1 pewna kobieta, którπ kocha John
G2 pewien mÍøczyzna, który zna Johna i którego zna kobieta G1


